

## Maple - Seminar 3

**EXAMPLE 1:** How many real solutions does the equation  $x = 2000 \sin(x)$  have?

```
> restart;
```

```
> eq:=x=2000*sin(x);
```

```
eq := x = 2000 sin(x)
```


```
> allvalues(solve(eq,x));
```

```
RootOf(-_Z + 2000 sin(_Z), 0.)
```

```
> eq1:=x/2000=sin(x);
```

```
eq1 :=  $\frac{x}{2000} = \sin(x)$ 
```

```
> plot({lhs(eq1),rhs(eq1)},x=-100..100);
```



```
> trunc(2000/Pi);
```

```
636
```


```
> evalf(frac(2000/Pi));
```

```
0.6197722
```

```
> 2*(trunc(2000/Pi)+2)-1;
```

```
1275
```

**EXAMPLE 2:** Rotate the finite region bounded by the curves  $y = x^2$  and  $y = x$  around the line  $y = x$  (See the figure below).



What is the volume of the resulting solid?

What is its surface area?

Try to represent the corresponding solid of revolution using the means of Maple software.

```
> restart;
> DistFromLine:=proc(A,r)
>     local a,b,c,d;
>     a:=coeff(r,x): b:=coeff(r,y): c:=r-a*x-b*y;
>     d:=abs(A[1]*a+A[2]*b+c)/sqrt(a^2+b^2);
>     d;
> end:
> f:=x->x^2;
```

$$f := x \rightarrow x^2$$

**I.** The volume is obtained by evaluation of the formula  $V = \int_0^R \pi p^2 dr$ , where  $p$  is the

perpendicular distance of a point  $[x, x^2]$  on the parabola from the line  $x - y = 0$ ,  $r$  is a distance of the foot of the perpendicular from the origin. After substitution of  $p$  and  $r$  in the formula we will integrate with respect to  $x$  between  $x=0$  and  $x=1$ .

The surface area is obtained by integrating the formula  $S = \int_{s_0}^{s_1} 2 \pi p ds$ , where  $ds$  is the element

of arc-length of the parabola,  $ds = \sqrt{1 + \left(\frac{d}{dx} f(x)\right)^2} dx$ . After substitution we will integrate with respect to  $x$  between  $x=0$  and  $x=1$  again.

```
> p:=DistFromLine([x,f(x)],x-y);
```

$$p := \frac{1}{2} |-x + x^2| \sqrt{2}$$

```
> assume(x,real);
```

```
> r:=simplify(sqrt((x^2+f(x)^2)-p^2));
```

$$r := \frac{1}{2} \sqrt{2} |x~ (1 + x~)|$$

```
> diff(r,x);
```

$$\frac{1}{2} \sqrt{2} \text{signum}(x~ (1 + x~)) (2 x~ + 1)$$

```
> V:=int(Pi*p^2*diff(r,x),x=0..1); evalf(V);
```

$$V := \frac{\pi \sqrt{2}}{60}$$

0.07404804898

```
> S:=int(2*Pi*p*sqrt(1+4*x^2),x=0..1); evalf(S);
```

$$S := -\frac{\pi \sqrt{2}}{12} + \frac{13 \pi \sqrt{2} \sqrt{5}}{96} - \frac{1}{64} \pi \sqrt{2} \ln(-2 + \sqrt{5})$$

1.075285824

II. We use the frame transformation - revolution around the origin to make identical a new  $x$  axis and the line  $x - y = 0$

```
> restart;
```

```
> f:=x->x^2;
```

$$f := x \rightarrow x^2$$

Variables  $x, y$  in the formula  $y=f(x)$  are substituted by the expressions  $\frac{\sqrt{2}(x-y)}{2}$ ,  $\frac{\sqrt{2}(x+y)}{2}$

respectively.

```
> fTr:=simplify(subs({x=sqrt(2)/2*(x-y),y=sqrt(2)/2*(x+y)},y=f(x)));
```

$$fTr := \frac{\sqrt{2}(x+y)}{2} = \frac{(x-y)^2}{2}$$

```
> ySol:=solve(fTr,y);
```

$$ySol := \frac{\sqrt{2}}{2} + x + \frac{\sqrt{2+8\sqrt{2}x}}{2}, \frac{\sqrt{2}}{2} + x - \frac{\sqrt{2+8\sqrt{2}x}}{2}$$


Graph of the  $f(x)$  function is split up into two functions  $F1(x)$ ,  $F2(x)$  after the revolution. Next we are interested in the  $F2(x)$  function (see the green curve in figure).

```
> F1:=unapply(ySol[1],x); F2:=unapply(ySol[2],x);
```

$$F1 := x \rightarrow \frac{\sqrt{2}}{2} + x + \frac{1}{2}\sqrt{2+8\sqrt{2}x}$$

$$F2 := x \rightarrow \frac{\sqrt{2}}{2} + x - \frac{1}{2}\sqrt{2+8\sqrt{2}x}$$

```
> plot({F1(x),F2(x)},x=-1..2,scaling=constrained);
```



$$V = \int_0^{\sqrt{2}} \pi F2(x)^2 dx$$

```
> V:=int(Pi*F2(x)^2,x=0..sqrt(2)); evalf(V);
```

$$V := \frac{\sqrt{2} \pi}{60}$$

0.07404804898

$$S = \int_0^{\sqrt{2}} 2\pi F_2(x) \sqrt{1 + \left(\frac{d}{dx} F_2(x)\right)^2} dx$$


```
> S:=int(2*Pi*F2(x)*sqrt(1+diff(F2(x),x)^2),x=0..sqrt(2));
evalf(S);
```

$$S := \frac{\sqrt{2}\pi}{12} - \frac{13\sqrt{2}\pi\sqrt{5}}{96} + \frac{1}{64}\sqrt{2}\pi \ln(-2 + \sqrt{5})$$


-1.075285824

To plot the rotational solid we use a parametric representation of its surface


```
>
plot3d([x,F2(x)*cos(u),F2(x)*sin(u)],x=0..sqrt(2),u=0..2*Pi,scaling=constrained);
```



**EXAMPLE 3:** If three points are chosen at random (and uniformly) on the circumference of a circle of radius 1, what is the expected value of the resulting triangle?



```
> restart;
> Circle:=plot([cos(t),sin(t),t=0..2*Pi]):
> C:=[1,0]:A:=[cos(alpha),sin(alpha)]:
> B:=[cos(beta),sin(beta)]:
>
> Triangle:=plots[polygonplot]([[1,0],[cos(3*Pi/4),sin(3*Pi/4)],[cos(5*Pi/3),sin(5*Pi/3)]],color=red):
>
> A:=plots[textplot]([cos(3*Pi/4),sin(3*Pi/4),"A"],align={ABOVE,LEFT}):
> B:=plots[textplot]([cos(5*Pi/3),sin(5*Pi/3),"B"],align={BELOW,RIGHT}):
> C:=plots[textplot]([1,0,"C"],align={ABOVE,RIGHT}):
> plots[display](Circle,Triangle,A,B,C);
```



Area of the triangle ABC ( $A = [\cos(\alpha), \sin(\alpha)]$ ,  $B = [\cos(\beta), \sin(\beta)]$ ,  $C$  is fixed to  $[1,0]$ ) is

$$\frac{\sin(\alpha) + \sin(\beta - \alpha) - \sin(\beta)}{2}$$

```
> Area:=simplify(1/2*sin(alpha)+1/2*sin(beta-alpha)-
1/2*sin(beta), trig);
```

$$Area := \frac{1}{2} \sin(\alpha) + \frac{1}{2} \sin(\beta - \alpha) - \frac{1}{2} \sin(\beta)$$

$$IntBeta = \int_0^{\alpha} -Area \, d\beta + \int_{\alpha}^{2\pi} Area \, d\beta$$

```
> IntBeta:=int(-Area,beta=0..alpha)+int(Area,beta=alpha..2*Pi);
IntBeta := -2 cos(alpha) - sin(alpha) alpha + 2 + sin(alpha) pi
```

$$IntAlpha = \int_0^{2\pi} IntBeta \, d\alpha$$

```
> IntAlpha:=int(IntBeta,alpha=0..2*Pi);
IntAlpha := 6 pi
```

$$Average = \frac{IntAlpha}{(2\pi - 0)^2}$$

```
> Av:=IntAlpha/(2*Pi)^2; evalf(Av);
```

$$Av := \frac{3}{2\pi}$$

0.4774648292

We can perform a computer simulation. Here it is for 100 random triangles.

```
> Suma:=0; n:=0;
```

Suma := 0

n := 0

```
> for i from 1 to 100 do
```

```
> RandAngle:=rand(0..360):
```

```
> a:=RandAngle()*Pi/180: b:=RandAngle()*Pi/180:
```

```
c:=RandAngle()*Pi/180:
```

```
>
```

```
Area[i]:=abs(evalf(1/2*linalg[det](linalg[matrix]([[cos(a),sin(a)
,1],[cos(b),sin(b),1],[cos(c),sin(c),1]]))))):
```

```
> Suma:=Suma+Area[i];
```

```
> od:
```


```
> Suma/100;
```

0.4633119583

```
> data:=[seq(Area[i],i=1..100)]:
```

```
> with(stats[statplots]):
```

```
histogram(data, color=cyan,area=count);
```



**PROBLEM 1:** If  $0 < a < 1$ , then the equation  $a^x = x$  has a unique solution, since the function on the left is strictly decreasing while the function on the right is strictly increasing.

For which value of  $a > 1$  does the equation  $a^x = x$  have a unique solution?

```
> restart;
```

```
> f:=x->a^x-x;
```

$$f := x \rightarrow a^x - x$$

```
> plots[interactive]();
```

```
> D(f)(x); (D@@2)(f)(x);
```

$$a^x \ln(a) - 1$$

$$a^x \ln(a)^2$$

```
> AbsMin:=simplify(solve(D(f)(x)=0,x));
```

$$AbsMin := -\frac{\ln(\ln(a))}{\ln(a)}$$

```
> UniqueSol:=simplify(subs(x=AbsMin,f(x)));
```

$$UniqueSol := \frac{1 + \ln(\ln(a))}{\ln(a)}$$

```
> UniqueSol_a:=solve(UniqueSol=0,a);
```


$$UniqueSol_a := e^{\left(\frac{1}{e}\right)}$$



**PROBLEM 2:** Consider the cup defined by the equation  $z = x^2 + y^2$  with  $z$  between 0 and 4. The cup is filled with water and slowly tipped. To the nearest degree, at what angle of tilt will half of the water be left in the cup?

Illustration:

```
> restart;
> Paraboloid:=plot3d([x*cos(u),x*sin(u),x^2],x=-
2..2,u=0..2*Pi,scaling=constrained):
> Plane:=plot3d([x,y,0.5*x+2.5],x=-4..4,y=-4..4):
> plots[display](Paraboloid,Plane);
```



Solution:

$$V = \int_{a-2}^2 \int_{y^2}^{ay+b} \int_{-\sqrt{ay+b-y^2}}^{\sqrt{ay+b-y^2}} 1 \, dx \, dz \, dy$$

$$b = 4 - 2a$$

```
> restart;
> yr:=solve(y^2=a*y+4-2*a,y);
                                     yr := 2, a - 2
> V:=int(int(int(1,x=-sqrt(a*y+4-2*a-y^2)..sqrt(a*y+4-2*a-
y^2)),z=y^2..a*y+4-2*a),y=yr[2]..yr[1],'AllSolutions');
```

$$V := \begin{cases} -12 \pi a + 12 \pi - \frac{3}{4} \pi a^3 + \frac{9}{2} \pi a^2 + \frac{3}{64} a^4 \pi & a \leq 4 \\ -12 \pi + 12 \pi a - \frac{9}{2} \pi a^2 + \frac{3}{4} \pi a^3 - \frac{3}{64} a^4 \pi & 4 < a \end{cases}$$

> **V1:=V assuming a<=4;**

$$V1 := -12 \pi a + 12 \pi - \frac{3}{4} \pi a^3 + \frac{9}{2} \pi a^2 + \frac{3}{64} a^4 \pi$$

> **V2:=simplify(factor(V1));**

$$V2 := \frac{3 \pi (a - 4)^4}{64}$$

> **as:=fsolve(V2-4\*Pi,a);**

$$as := 0.9606572575, 7.039342730$$

> **a:=as[1]; b:=4-2\*a;**

$$a := 0.9606572575$$

$$b := 2.078685485$$

> **V1:=int(int(int(1,x=-sqrt(a\*y+4-2\*a-y^2)..sqrt(a\*y+4-2\*a-y^2)),z=y^2..a\*y+4-2\*a),y=a-2..2,'AllSolutions');**

$$V1 := 12.56637061$$

> **evalf(V1/(4\*Pi));**

$$0.9999999992$$


> **angle:=evalf(convert(arctan(a),degrees));**

$$angle := 43.85045149 \text{ degrees}$$

> **Par:=plot3d([x\*cos(u),x\*sin(u),x^2],x=-2..2,u=0..2\*Pi,scaling=constrained):**

> **Plane:=plot3d([x,y,a\*y+b],x=-4..4,y=-4..4):**

> **plots[display](Par,Plane,axes=normal);**



```
> a:=as[2]; b:=4-2*a;
```

```
a := 7.039342730
```

```
b := -10.07868546
```

```
> V1:=int(int(int(1,x=-sqrt(a*y+4-2*a-y^2)..sqrt(a*y+4-2*a-y^2)),z=y^2..a*y+4-2*a),y=a-2..2,'AllSolutions');
```

```
V1 := -12.56637026 - 0.90707735 10-8 I
```

```
> evalf(V1/(4*Pi));
```

```
-0.9999999715 - 0.7218292199 10-9 I
```

```
> Par:=plot3d([x*cos(u),x*sin(u),x^2],x=-2..2,u=0..2*Pi,scaling=constrained):
```

```
> Plane:=plot3d([x,y,a*y+b],x=-4..4,y=-4..4):
```

```
> plots[display](Par,Plane):
```

```
>
```

```
>
```